Paper HS-53

Creating suicide attempt/intentional self-harm episodes using administrative billing data

Jamie Kammer, Mahfuza Rahman, and Qingxian Chen PSYCKES, Office of Mental Health, Albany, NY

Abstract

In identifying the services that individuals received in the time prior to and following a suicide attempt or intentional self-harm episode, it is important to separate out those services that occurred as part of treatment during the emergency room or inpatient visit (including transfers) from services outside of the episode. Specifically, it is useful to understand the service utilization patterns individuals experienced directly prior to a suicide attempt, whether individuals were engaged in outpatient care immediately following emergency room or inpatient treatment, and whether individuals experienced separate suicide attempts outside of an index episode. These insights can inform public health efforts to prevent and treat suicide attempts and intentional self-harm. There are several considerations to take into account when using administrative billing data for public health research and there is a need for standardizing methods around identifying episodes carefully. This paper describes one method for linking together rows of administrative billing data into continuous suicide attempt/intentional selfharm episode(s) with begin and end dates as a first step in service utilization analyses.

Introduction

Suicide attempts and intentional self-harm are common and pervasive public health concerns¹. Administrative billing data, like Medicaid, can be used to identify suicide attempts and intentional selfharm that have received medical attention in order to inform prevention and intervention efforts. However, it can be challenging to disentangle which services were part of an incident suicide attempt or intentional self-harm episode, as each row of billing data pertaining to the episode may not contain a diagnosis code specifically identifying suicide attempt or intentional self-harm. Diagnoses related to the specific injury incurred or mental health diagnoses may be used on bills that are part of the same episode of care. It can also be difficult to discern intentional self-harm from unintentional self-harm (especially those episodes of self-harm involving overdose of illegal drugs), highlighting the importance of carefully deciding which service settings to include in addition to considering diagnoses. Leaving out diagnoses received in substance abuse treatment settings may be helpful in reducing the number of false-positive cases identified. We used the following SAS code to create a cohort of individuals with suicide attempt or intentional self-harm diagnoses and select an index episode.

Selecting the initial data

Because not all services that are part of an episode of self-harm will contain suicide attempt or intentional self-harm diagnoses, the first step in creating this dataset will be creating a list of all individuals with any suicide attempt or intentional self-harm diagnosis code in any diagnosis field during the period of observation. In this step, we excluded those diagnoses with labels that included the terms "subsequent encounter" or "sequela" so that we would identify incident self-harm and suicide attempts only. We then identified all emergency department and inpatient records for all individuals on this list regardless of diagnoses received in these settings, creating a new variable indicating whether the record is for an inpatient or emergency department visit. Here we excluded inpatient rehab and other substance abuse service records in order to reduce the number of records with intentional self-harm diagnoses that were actually unintentional overdoses. We restricted inpatient visits to those with a date of admission that occurred during the period of observation and restricted emergency department visits on date of service. In addition to all emergency department and inpatient records, we also selected

outpatient and transportation records with a date of service during the period of observation that contained a suicide attempt or intentional self-harm diagnosis in any field. The outpatient and transportation data were saved in a separate dataset from the emergency department and inpatient records. The final selected variables include the client identification variable, all diagnosis fields, service date, admission date, discharge date, and service end date.

Table 1. Clients with any suiciderelated diagnosis during year

Client ID	
1	
2	
3	

Table 2. Emergency room and inpatient data

Client						
ID	ER/IP	Date of Service	Admission	Discharge	Diag Code 1	Diag Code 2
1	ER	2/1/2016	-	-	T370X2A	F3132
1	IP	2/2/2016	2/2/2016	2/4/2016	T1491	
1	IP	3/1/2016	3/1/2016	3/5/2016	F3132	
2	ER	2/16/2016	-	-	T1491	
2	ER	2/17/2016	-	-	F0632	
3	IP	1/1/2016	1/1/2016	1/3/2016	T1491	
***		the second second state of the second s	and the state of the second second			

*includes all ER/IP visits among individuals with any suicide diagnosis

Data cleaning

After the necessary service records are selected, the next step is cleaning inpatient record dates. In cases where the date of service was prior to the admission date, we replaced the admission date with the date of service. If the date of admission was missing we also replaced this with the date of service. If discharge date was missing this was replaced with the service end date.

Identifying rows with suicide attempt or self-harm diagnoses

The next step is creating a dichotomous variable identifying whether the row contains a suicide attempt or intentional self-harm diagnosis code. In this case, we used the same list of diagnoses that we used to identify the initial cohort of individuals with a suicide attempt or intentional self-harm diagnosis.

```
proc sql;
```

```
create table ER_INPT_DX as
select distinct *, case when
  (diag_code1 in (select distinct diag_cd from suicide_dx)
or diag_code2 in (select distinct diag_cd from suicide_dx)
or diag_code3 in (select distinct diag_cd from suicide_dx)
or diag_code4 in (select distinct diag_cd from suicide_dx)
or diag_code5 in (select distinct diag_cd from suicide_dx)
or diag_code6 in (select distinct diag_cd from suicide_dx)
or diag_code6 in (select distinct diag_cd from suicide_dx)
or diag_code7 in (select distinct diag_cd from suicide_dx)
or diag_code8 in (select distinct diag_cd from suicide_dx)
or diag_code8 in (select distinct diag_cd from suicide_dx)
or diag_code9 in (select distinct diag_cd from suicide_dx)
```

```
or diag_code10 in (select distinct diag_cd from suicide_dx))
then 1 else 0 end as suicide_dx
from ER_INPT;
quit;
```

Client							
ID	ER/IP	Date of Service	Admission	Discharge	Diag Code 1	Diag Code 2	Suicide DX
1	ER	2/1/2016	-	-	T370X2A	F3132	1
1	IP	2/2/2016	2/2/2016	2/4/2016	T1491		1
1	IP	3/1/2016	3/1/2016	3/5/2016	F3132		0
2	ER	2/16/2016	-	-	T1491		1
2	ER	2/17/2016	-	-	F0632		0
3	IP	1/1/2016	1/1/2016	1/3/2016	T1491		1

Table 3. Emergency room and inpatient data with suicide diagnosis indicator variable

Stringing together services that are part of the same episode

In the next steps, we assign begin and end dates to each row of the emergency room and inpatient data. For inpatient records, these will reflect the admission and discharge dates; in the case of emergency room records the begin and end dates will both be the date of service.

```
data ER_INPT_DX2;
  set ER_INPT_DX;
  format begin_dt end_dt mmddyy10.;
  if find (ER_IP, 'inpatient','i') ge 1 then
    do;
       begin_dt = date_admission;
       end_dt = date_discharge;
    end;
  else
    do;
       begin_dt = date_of_service;
       end_dt = date_of_service;
       end;
  run;
```

Table 4.	Assigning	begin	and	end	dates

Client ID	ER/IP	Date of Service	Admission	Discharge	Suicide DX	Begin Date	End Date
1	ER	2/1/2016	-	-	1	2/1/2016	2/1/2016
1	IP	2/2/2016	2/2/2016	2/4/2016	1	2/2/2016	2/4/2016
1	IP	3/1/2016	3/1/2016	3/5/2016	0	3/1/2016	3/5/2016
2	ER	2/16/2016	-	-	1	2/16/2016	2/16/2016
2	ER	2/17/2016	-	-	0	2/17/2016	2/17/2016
3	IP	1/1/2016	1/1/2016	1/3/2016	1	1/1/2016	1/3/2016

The following code removes any records that are missing a begin or end date and selects and orders those variables that we need in subsequent steps.

proc sql; create table ER INPT DX3 as

```
select distinct client_id, suicide_dx, begin_dt, end_dt
from ER_INPT_DX2
where begin_dt ne . and end_dt ne .
order by client_id, begin_dt, end_dt;
quit;
```

Table 5.	Begin	and	end	dates
----------	-------	-----	-----	-------

Client ID	Suicide DX	Begin Date	End Date
1	1	2/1/2016	2/1/2016
1	1	2/2/2016	2/4/2016
1	0	3/1/2016	3/5/2016
2	1	2/16/2016	2/16/2016
2	0	2/17/2016	2/17/2016
3	1	1/1/2016	1/3/2016

The next step links together continuous services that are part of the same episode. If services were received on the same day or within one day of each other, these were counted as part of one continuous episode. This can also be changed to use less strict criteria for discerning one episode from the next.

```
data ER INPT DX4;
set ER INPT DX3;
by client id begin dt end dt;
retain b e seqn;
format b e mmddyy10.;
if first.client id then do;
b=begin dt;
e=end dt;
seqn=1;
end;
if (begin dt<=e+1) then do;</pre>
e=max(e, end dt);
end;
else do;
seqn=seqn+1;
b=begin dt;
e=end dt;
end;
run;
proc sort data= ER INPT DX4;
by client id seqn begin dt end dt;
run;
```

Client ID	Suicide DX	Begin Date	End Date	Seqn
1	1	2/1/2016	2/1/2016	1
1	1	2/2/2016	2/4/2016	1
1	0	3/1/2016	3/5/2016	2
2	1	2/16/2016	2/16/2016	1
2	0	2/17/2016	2/17/2016	1
3	1	1/1/2016	1/3/2016	1

Table 6. Putting services together into episodes of care

In order to identify additional suicide-related episodes where the self-harm or suicide attempt diagnoses were received in transportation or outpatient settings only, we joined the emergency room/inpatient dataset with outpatient and transportation records with a suicide attempt or self-harm diagnosis. In these cases, the suicide diagnosis was received during an emergency department/inpatient episode or within one day prior.

```
proc sql;
```

```
create table ER_INPT_DX5 as
select distinct a.*, suicide_dx2
from ER_INPT_DX4 a left join transport_outpt b
on a.client_id=b.client_id
and datepart(date_of_service)>=begin_dt-1
and datepart(date_of_service)<=end_dt;
quit;
```

Client ID	Suicide DX	Begin Date	End Date	Seqn	Suicide DX2
1	1	2/1/2016	2/1/2016	1	1
1	1	2/2/2016	2/4/2016	1	0
1	0	3/1/2016	3/5/2016	2	0
2	1	2/16/2016	2/16/2016	1	0
2	0	2/17/2016	2/17/2016	1	0
3	1	1/1/2016	1/3/2016	1	0

Table 7. Adding transportation and outpatient records with suicide diagnosis

The next steps include identifying rows with any suicide-related diagnoses and taking the maximum value of the final suicide diagnosis indicator variable to identify which emergency department/inpatient episodes can be considered part of a suicide attempt or intentional self-harm. We then create a dataset that contains only those emergency room and/or inpatient episodes that are considered part of treatment for a suicide attempt or intentional self-harm. Because individuals may have more than one episode, we took the most recent during the observation period to identify the index episode begin and end dates.

```
data ER_INPT_DX6;
set ER_INPT_DX5;
if suicide_dx=1 or suicide_dx2=1 then suicide_dx3=1;
else suicide_dx3=0;
run;
```

```
proc sql;
create table ER INPT DX7 as
select distinct client id, max(suicide dx3) as suicide dx,
begin dt, end dt, b, e, seqn,
from ER INPT DX6
group by client id, seqn
order by client id, seqn, begin dt, end dt;
quit;
data ER INPT DX8;
set ER INPT DX7;
by client id seqn begin dt end dt;
if last.seqn;
begin dt=b;
end dt=e;
drop b e;
run;
proc sql;
create table ER INPT DX9 as
select distinct *
from ER INPT DX8
where suicide dx=1
order by client id, seqn;
quit;
data ER INPT DX10;
set ER INPT DX9;
by client id;
if last.client id;
drop seqn;
run;
```

Results

Below is an example of what the final dataset will look like. This dataset can then be used to assess service utilization patterns before and after emergency department and/or inpatient treatment for an index episode related to a suicide attempt or intentional self-harm.

	asel			
Client ID	Suicide_DX	Begin_DT	End_DT	
1	1	2/1/2016	2/4/2016	
2	1	2/16/2016	2/17/2016	
3	1	1/1/2016	1/3/2016	

Table 8. Final dataset

In running this code on the New York State Medicaid data, 10,701 episodes of suicide attempt or intentional self-harm treated in an emergency room or inpatient setting were identified between 11/1/2015 and 10/31/2016. Of these, 6,699 (63%) included an inpatient stay. Median episode length was 3 days.

Conclusions

This code can be used to identify episodes of care for suicide attempts and intentional self-harm that were treated in an emergency department or inpatient setting. This method can be used to discern which services occurred as part of the index episode of care using the episode's begin and end dates. It

is important to note that the final dataset does not differentiate suicide attempts from self-harm not intended to lead to death. Also, we restricted to those episodes treated in an emergency room or inpatient setting only, which excludes self-harm treated in an outpatient setting or left untreated.

Acknowledgements

This work was supported, in part, by the New York State Office of Mental Health and NIMH grant R01 MH112139-01. The authors would like to thank the other members of the Data Analysis Work Group for their input on the developed definition: Molly Finnerty, Emily Leckman-Westin, Barbara Stanley, Prabu Vasan, Anni Kramer, Deborah Layman, Christa Labouliere, Hanga Ganfalvy, Greg Brown, Kelly Green, Paige Allen, Yvonne Noriega. Views presented here do not necessarily reflect the views of NYS OMH.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. [®] indicates USA registration.

References:

1. WISQARS. Retrieved from <u>https://webappa.cdc.gov/sasweb/ncipc/nfirates.html</u> on 2/21/2018.